

## Chapter 6

### External Economies and Monopolistic Competition

Now we continue our study of increasing returns to scale, turning to models based on external economies of scale and monopolistic-competition models. I have grouped these two together in one chapter since there are close technical similarities between them. A formal demonstration of this equivalence is found in Markusen (CJE vol. 23, 1990, 495-508.).

These models require a few tricks to code in MPS/GE, because MPS/GE requires all activities to have constant returns to scale. Because of this, I want to present MCP versions of the models first, in order to show clearly what equations and inequalities are being solved. Then I present the MPS/GE versions and show the tricks. In very simple models such as the ones presented here, it is generally easiest to stick with the MCP solver and not use the higher-level language. But in more complicated higher-dimension models, it is best to learn the tricks and use MPS/GE.

Models are as follows:

M61-MCP.GMS	Closed economy external economies of scale, MCP version
M61-MPS.GMS	Closed economy external economies of scale, MPS/GE version
M62-MCP.GMS	Closed economy large-group monopolistic competition, MCP version
M62-MPS.GMS	Closed economy large-group monopolistic competition, MPS/GE version
M63-MCP.GMS	Two-country model with large-group monopolistic competition, MCP version
M63-MPS.GMS	Two-country model with large-group monopolistic competition, MPS/GE version

Model 61-MCP      Closed economy external economies of scale, MCP version

The theoretical foundations of this model go back to the 1960s with work by Kemp and Melvin. The idea is that the production function for an individual firm is given by

$$(1) \quad X_i = (X^\beta)F(V_i) \quad 0 < \beta < 1$$

where  $X_i$  is firm  $i$ 's output,  $X$  is the sum of all firms' output (industry output),  $F$  is a function with constant returns to scale, and  $V_i$  is the firm's vector of inputs.

The parameter  $\beta$  reflects industry level scale economies, with  $\beta = 0$  being the special case of constant returns to scale.

By virtue of the fact that  $F(\cdot)$  has constant returns to scale and is identical for all firms, total industry output can then be written as:

$$(2) \quad X = (X^\beta)F(V) \quad \text{or} \quad X = F(V)^{1/(1-\beta)}$$

so  $X$  industry output is homogeneous of degree  $1/(1 - \beta) > 1$  in primary factor inputs.

To make the point in the simplest fashion, assume that there is just a single factor  $L$  with a price  $w$ .  $p$  denotes the price of  $X$ . Each individual firm views total industry output as constant, and thus each firm views itself as producing with constant returns to scale. The firm's optimization condition equates the value of the (private) marginal product of labor to the wage rate.

$$(3) \quad pX^\beta \frac{dX_i}{dL_i} = w \quad p = X^{-\beta} \left[ w \frac{dL_i}{dX_i} \right] \equiv X^{-\beta} c(w)$$

where  $c(w)$  is the cost of one "factor bundle", the cost of labor needed to produce  $F(L) = 1$ . That is,  $c(w)$  is the unit cost function associated with the constant-returns function  $F(L)$ . This then becomes the pricing equation or zero profit condition for the  $X$  sector in the MCP.

The other thing that we need for the market-clearing condition for labor in the MCP is the total demand for labor in the  $X$  sector. This is given by the amount of labor needed for  $F(L) = 1$  times  $F(L)$ . The unit requirement is given as usual by Shephard's lemma, the derivative of the unit cost function. So using (1), the demand for labor in the  $X$  sector is given by

$$(4) \quad c'(w)F(L) = c'(w)X^{1-\beta}$$

The demand for labor is concave in output, the dual of increasing returns in production.

Here is the data matrix for the model, the same one that we have used before:

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PU			200	-200
PW	-40	-60		100
PZ	-60	-40		100

We will assume that the production functions are Cobb-Douglas. Units will be chosen such that  $X = 1$  initially, which then greatly simplifies the calibration of the data to the model. The letter “B” is used in place of beta. For some historical reasons now forgotten, PU now denotes the price of utility (welfare), PW is the price of unskilled labor, and the other factor is now called skilled labor, with a price of PZ.

The following block of code is inserted in order to suppress a lot of output in the listing file and you can certainly ignore it for now. But it is valuable in a program with many solve statements.

```
OPTION LIMROW=0;
OPTION LIMCOL=0;
$OFFSYMLIST OFFSYMREF OFFFUELLIST OFFFUELXREF
```

Here is the model.

\$TITLE: Model M61-MCP: External Economies of Scale, uses MCP

\$ONTEXT

The model is based on the benchmark social accounts for model M1-1:

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PU			200	-200
PW	-40	-60		100
PZ	-60	-40		100

\$OFFTEXT

PARAMETERS

B

ENDOWS

ENDOWL

MODELSTAT;

ENDOWS = 100;

ENDOWL = 100;

B = 0.2;

POSITIVE VARIABLES

X

Y

W

PX

PY

PU

PZ

PW

CONS;

EQUATIONS

DX Demand for X

DY Demand for Y

DW Demand for W

PRICEX MR = MC in X

PRICEY Zero profit condition for Y (PY = MC)

PRICEW Zero profit condition for W

SKLAB Supply-demand balance for skilled labor

UNLAB        Supply-demand balance for unskilled labor  
 INCOME      National income;

PRICEX..     $(PW^{**0.40}) * (PZ^{**0.60}) / (X^{**B}) =G= PX;$

PRICEY..     $(PW^{**0.60}) * (PZ^{**0.40}) =G= PY;$

PRICEW..     $(PX^{**0.50}) * (PY^{**0.50}) =G= PU;$

DX..         $X*100 =E= CONS / (2*PX);$

DY..         $Y*100 =E= CONS / (2*PY);$

DW..         $200*W =E= CONS / PU;$

SKLAB..     $ENDOWS =E= 0.40 * (PW^{**0.60}) * (PZ^{** (0.40-1)}) * Y*100$   
 $+ 0.60 * (PW^{**0.40}) * (PZ^{** (0.60-1)}) * (X^{** (1-B)}) * 100;$

UNLAB..     $ENDOWL =E= 0.60 * (PW^{** (0.60-1)}) * (PZ^{**0.40}) * Y*100$   
 $+ 0.40 * (PW^{** (0.40-1)}) * (PZ^{**0.60}) * (X^{** (1-B)}) * 100;$

INCOME..     $CONS =E= PZ*ENDOWS + PW*ENDOWL;$

MODEL M61 /DX.PX, DY.PY, DW.PU, PRICEX.X, PRICEY.Y, PRICEW.W,  
 SKLAB.PZ, UNLAB.PW, INCOME.CONS/;

OPTION MCP=MILES;

OPTION LIMROW=0;

OPTION LIMCOL=0;

\$OFFSYMLIST OFFSYMXREF OFFUELLIST OFFFUELXREF

CONS.L = 200;

X.L = 1;

Y.L = 1;

W.L = 1;

PX.L = 1;

PY.L = 1;

PZ.L = 1;

PW.L = 1;

PU.L = 1;

PY.FX = 1;

```
SOLVE M61 USING MCP;  
MODELSTAT = M61.MODELSTAT - 1.;
```

```
DISPLAY MODELSTAT;
```

```
*           Counterfactual: expand the size of the economy
```

```
ENDOWS = 200;  
ENDOWL = 200;
```

```
SOLVE M61 USING MCP;
```

```
*           Counterfactual: contract the size of the economy
```

```
ENDOWS = 80;  
ENDOWL = 80.
```

```
SOLVE M61 USING MCP;
```

The counterfactual-experiments involve changing the size of the economy. The first experiment doubles the size of the economy. This increases the welfare index from 1.000 to 2.181, more than double its initial value representing the scale economies in production.

Model 61-MPS      Closed economy external economies of scale, MPS/GE version

The difficulty for translating this model into MPS/GE is that this higher-level language requires constant returns to scale in all activities so that it can generate cost functions and factor demands in a standardized way. There are a couple of tricks that get around this. They may seem awkward and not worth the bother in such a small model, but in big models with many sectors, factors, or countries it is well worth learning the tricks.

How do we model this given that MPS/GE requires constant returns to scale? We will use a trick, in which the X industry produces  $X_0 = F(V_0)$  so that the X sector output is now the constant returns function F using factor bundles. Then we will "give" the consumer an additional amount equal to the "true" industry output minus the output from factor bundles.

$$(5) \quad X_1 = X_0^{1/(1-\beta)} - X_0$$

Then the consumer actually receives

$$(6) \quad X = X_1 + X_0 = F(V_0)^{1/(1-\beta)}.$$

The value of  $X_1$  is determined by the auxiliary variable XQADJ (X quantity adjustment =  $X_1$ ) in the program that follows. It is important to note that the value of X listed in the output file of the program is what is called  $X_0$ , not the true output of X.

But now we have an imbalance in that the value of X received by the consumer is more than the value of X produced by the firms. So we will subsidize X production so that the value of payments received by the firm for  $X_0$  is equal to the value of  $X = X_1 + X_0$  consumed by the consumer. Let q denote the consumer price and p the producer price of X. Let q be the tax base, so  $q(1-s) = p$ . For payments to balance, we need

$$(7) \quad q(X_0^{1/(1-\beta)}) = pX_0 \quad \text{or} \quad (1-s) = X_0^{\beta/(1-\beta)}$$

$$(8) \quad s = X_0^{\beta/(1-\beta)} - 1 \quad \text{or} \quad s = XQADJ/X_0$$

s is determined in the model to follow by the auxiliary variable XPADJ.

The model is calibrated so that all activity levels are one initially, implying that the initial values of XQADJ and XPADJ are zero initially.  $\beta = .2$ ,  $\beta/(1-\beta) = .25$

Counterfactual experiments change the size of the economy. Notice the consequences of the scale economies.

\$TITLE: Model M61-MPS: External Economies of Scale, MPS/GE version

\$ONTEXT

The model is based on the benchmark social accounts for model M1-1:

Markets	Production Sectors			Consumers
	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PU			200	-200
PW	-40	-60		100
PZ	-60	-40		100

\$OFFTEXT

PARAMETER

ENDOW Size index for the economy  
B External economies parameter;

ENDOW = 1;  
B = 0.2

\$ONTEXT

\$MODEL:M61

\$SECTORS:

X ! Activity level for sector X  
Y ! Activity level for sector Y  
W ! Activity level for sector W (Hicksian welfare index)

\$COMMODITIES:

PX ! Price index for commodity X  
PY ! Price index for commodity Y  
PW ! Price index for primary factor L  
PZ ! Price index for primary factor S  
PU ! Price index for welfare (expenditure function)

\$CONSUMERS:

CONS ! Income level for consumer CONS

\$AUXILIARY:

XQADJ ! Quantity adjustment (positive when X>1)

XPADJ ! X output subsidy rate (positive when  $X > 1$ )

\$PROD: X s:1  
 O:PX Q:100 A:CONS N:XPADJ M:-1  
 I:PW Q: 40  
 I:PZ Q: 60

\$PROD: Y s:1  
 O:PY Q:100  
 I:PW Q: 60  
 I:PZ Q: 40

\$PROD: W s:1  
 O: PU Q:200  
 I: PX Q:100  
 I: PY Q:100

\$DEMAND:CONS  
 D:PU Q:200  
 E:PW Q:(ENDOW\*100)  
 E:PZ Q:(ENDOW\*100)  
 E:PX Q:100 R:XQADJ

\$CONSTRAINT:XQADJ  
 $XQADJ = E = X^{**} (1 / (1 - B)) - X;$

\$CONSTRAINT:XPADJ  
 $XPADJ * X = E = XQADJ;$

\$OFFTEXT  
 \$SYSINCLUDE mpsgeset M61

\* Adjust bounds so that the auxiliary variables can take on  
 \* negative values:

XQADJ.LO = -INF;  
 XPADJ.LO = -INF;

\* Benchmark replication

\$INCLUDE M61.GEN  
 SOLVE M61 USING MCP;

\* Counterfactual: expand the size of the economy

```
ENDOW = 2;  
$INCLUDE M61.GEN  
SOLVE M61 USING MCP;
```

\* Counterfactual: contract the size of the economy

```
ENDOW = 0.8;  
$INCLUDE M61.GEN  
SOLVE M61 USING MCP;
```

Note again in comparing the results of the MCP and MPS/GE versions of the model that in the latter, the reported value of X denotes what we refer to above as  $X_0$ , rather than the “true” value of X industry output given by

$$X = X_0^{1/(1-\beta)}$$

which X is reported by the MCP solver.

Model 62-MCP Closed economy large-group monopolistic competition, MCP version

Now we turn to a favorite model used extensively in the new trade theory, the Dixit-Stiglitz model of monopolistic competition. Y will continue to be a competitive, constant-returns industry while X will consist of an endogenous number of differentiated varieties. Utility of the representative consumer in each country is Cobb-Douglas, and the symmetry of varieties within a group of goods allows us to write utility as follows ( $0 < \alpha < 1$ ).

$$(9) \quad U = X_c^\beta Y^{1-\beta}, \quad X_c \equiv \left[ \sum_i^N (X_i)^\alpha \right]^{\frac{1}{\alpha}}$$

where the number of varieties N is endogenous. This function permits the use of two-stage budgeting, in which the consumer in country i first allocates total income (M) between Y and  $X_c$ . Let  $X_c$  be as defined above, and let e denote the minimum cost of buying one unit of  $X_c$  at price p for the individual varieties (i.e., e is the unit expenditure function for  $X_c$ ). Y is numeraire. First-stage budgeting yields:

$$(10) \quad Y = (1-\beta)M \quad X_c = \beta M/e \quad e(p^k) = \min(X_i) \sum_i p X_i \quad st \quad X_c = 1$$

Let  $M_x = \beta M$  be the expenditure on X in aggregate. We need to solve for the demand for a given X variety, and for the price index e. The consumer's sub-problem maximizing the utility from X goods subject to an expenditure constraint (using  $\lambda$  as a Lagrangean multiplier) and first-order conditions are:

$$(11) \quad \max X_c = \left[ \sum X_i^\alpha \right]^{\frac{1}{\alpha}} + \lambda (M_x - \sum p_i X_i) \Rightarrow \frac{1}{\alpha} \left[ \sum X_i^\alpha \right]^{\frac{1}{\alpha} - 1} \alpha X_i^{\alpha-1} - \lambda p_i = 0$$

Let  $\sigma$  denote the elasticity of substitution among varieties. Dividing the first-order condition for variety i by the one for variety j,

$$(12) \quad \left[ \frac{X_i}{X_j} \right]^{\alpha-1} = \frac{p_i}{p_j} \quad \frac{X_i}{X_j} = \left[ \frac{p_i}{p_j} \right]^{\frac{1}{\alpha-1}} = \left[ \frac{p_i}{p_j} \right]^{-\sigma} \quad \text{since} \quad \sigma = \frac{1}{1-\alpha}$$

$$(13) \quad X_j = \left[ \frac{p_i}{p_j} \right]^\sigma X_i \quad p_j X_j = p_j p_j^{-\sigma} p_i^\sigma X_i \quad \sum p_j X_j = M_x = \left[ \sum p_j^{1-\sigma} \right] p_i^\sigma X_i$$

Inverting this last equation, we have the demand for an individual variety i:

$$(14) \quad X_i = p_i^{-\sigma} \left[ \sum p_j^{1-\sigma} \right]^{-1} M_x \quad \sigma = \frac{1}{1-\alpha}, \quad \alpha = \frac{\sigma-1}{\sigma}$$

Now we can use  $X_i$  to construct  $X_c$  and then solve for  $e$ , noting the relationship between  $\alpha$  and  $\sigma$ .

$$(15) \quad X_i^\alpha = X_i^{\frac{\sigma-1}{\sigma}} = p_i^{1-\sigma} \left[ \sum p_j^{1-\sigma} \right]^{\frac{1-\sigma}{\sigma}} M_x^\alpha$$

$$(16) \quad \sum X_i^\alpha = \left[ \sum p_i^{1-\sigma} \right] \left[ \sum p_j^{1-\sigma} \right]^{\frac{1-\sigma}{\sigma}} M_x^\alpha = \left[ \sum p_j^{1-\sigma} \right]^{\frac{1}{\sigma}} M_x^\alpha$$

$$(17) \quad X_c = \left[ \sum X_i^\alpha \right]^{\frac{1}{\alpha}} = \left[ \sum X_i^\alpha \right]^{\frac{\sigma}{\sigma-1}} = \left[ \sum p_j^{1-\sigma} \right]^{\frac{1}{\sigma-1}} M_x$$

$$(18) \quad e = \left[ \sum p_j^{1-\sigma} \right]^{\frac{1}{1-\sigma}}$$

Having derived  $e$ , we can then use equation (13) in (9) to get the demand for an individual variety.

$$(19) \quad X_i \equiv p_i^{-\sigma} e^{\sigma-1} M_x \quad \text{since} \quad e^{\sigma-1} = \left[ \sum p_j^{1-\sigma} \right]^{-1}$$

The usual assumption in “large-group” monopolistic competition is that there are many firms such that individual firms view  $e$  and  $M$  as constants. Thus the elasticity of demand for an individual variety is just  $\sigma$ . Equilibrium in the  $X$  sector involves two equations in two unknowns. The unknowns are  $X$ , output per variety and  $N$ , the numbers of varieties or firms. The two equations are the firm’s optimization condition, marginal revenue equals marginal cost, and the free-entry or zero profit condition, prices equals average cost. Let  $c(w,z)$  denote marginal cost where  $w$  and  $z$  are the prices of skilled and unskilled labor respectively. Let  $F(w,z)$  denote fixed costs per firm. The two equations are given by

$$(20) \quad p(1 - 1/\sigma) = c(w, z) \quad p = c(w, z) + F(w, z)/X$$

$$(21) \quad \frac{1}{(1 - 1/\sigma)} = 1 + \frac{F(w, z)}{c(w, z)X}$$

which simplifies to

$$(22) \quad X = (\sigma - 1) \frac{F(w, z)}{c(w, z)}$$

Another assumption that is typically made in the literature (often implicitly without realizing it) is that  $F$  and  $c$  have the same functional form, same factor intensities, etc. Under such assumptions, the right-hand side is a constant and does not depend on factor prices. I will use this assumption here. At initial factor prices,  $c = 1$  and  $F = 20$  in the MCP calibration, and so this ratio always equals 20 regardless of equilibrium factor prices. This is reflected in the zero-profit condition for  $X$  below: the equation ZEROP is (22) at  $c = 1$ .

The starting data matrix is exactly the same as that for model M61. The variables are:

#### POSITIVE VARIABLES

X	Output of an individual X variety
Y	Output of the Y industry
W	Welfare or utility
N	Number of varieties produced in equilibrium
E	Cost of producing one unit of $X_c$ (unit expenditure function for $X_c$ )
PX	Price of an individual X variety
PY	Price of Y
PZ	Price of skilled labor
PW	Price of unskilled labor
PU	Price of a unit of utility (the real consumer price index)
CONS;	Consumer income (M in the notation above)

The equations of the model are as follows, with complementary variables in parentheses.

#### EQUATIONS

ZEROP	Zero profits - free entry condition in X	(N)
PRICEY	Zero profit condition for Y	(Y)
PRICEW	Zero profit condition for W	(W)
PRICEX	MR = MC in X	(X)
INDEX	Price index for X sector goods	(E)
DX	Supply-demand balance for X	(PX)
DY	Supply-demand balance for Y	(PY)
DW	Supply-demand balance for utility W(welfare)	(PU)
SKLAB	Supply-demand balance for skilled labor	(PZ)
UNLAB	Supply-demand balance for unskilled labor	(PW)
INCOME	National income;	(CONS)

Here again we see the logic of associating zero profit or pricing equations with quantity

variables, and market clearing conditions with price.

One thing that is slightly misleading is to call the ZEROP equation “zero profits in X”. In the code to follow this is the equation for equilibrium output per firm in the X industry which is derived from using both the zero-profit condition and the pricing equation. This is not really necessary and we could just use the zero-profit condition here.

The code of the model follows:

\$TITLE: Model M62-MCP: Large-Group Monopolistic Competition: uses MCP

\$ONTEXT

Markets	Production Sectors				Consumers		ENTR
	XC	X	N	Y	W	CONS	
PX		100			-100		
CX	100	-100					
PY				100	-100		
PF			20				-20
PU					200	-200	
PW	-32		-8	-60		100	
PZ	-48		-12	-40		100	
MK	-20						20

\$OFFTEXT

PARAMETERS

EP

FC

ENDOWS

ENDOWL

WELFARE

MODELSTAT;

EP = 5;

FC = 20;

ENDOWS = 100;

ENDOWL = 100;

POSITIVE VARIABLES

X

Y

W

N

E

PX

PY

PZ

PW

PU

CONS;

## EQUATIONS

ZEROP Zero profits - free entry condition in X (associated with N)  
 PRICEY Zero profit condition for Y ( $PY = MC$ )  
 PRICEW Zero profit condition for W ( $PU = MC$  of utility)  
 PRICEX  $MR = MC$  in X (associated with X, output per firm)  
 INDEX Price index for X sector goods (unit expenditure function)  
 DX Supply-demand balance for X  
 DY Supply-demand balance for Y  
 DW Supply-demand balance for utility W (welfare)  
 SKLAB Supply-demand balance for skilled labor  
 UNLAB Supply-demand balance for unskilled labor  
 INCOME National income;

$$DX.. \quad X*80 \quad =E= \quad PX**(-EP)*(E**(EP-1))*CONS/2;$$

$$DY.. \quad Y*100 \quad =E= \quad CONS/(2*PY);$$

$$DW.. \quad 200*W \quad =E= \quad (1.25**0.5)*CONS/PU;$$

$$ZEROP.. \quad FC*(EP-1) \quad =G= \quad X*80;$$

$$PRICEX.. \quad (PW**0.4)*(PZ**0.6) \quad =G= \quad PX*(1-1/EP);$$

$$PRICEY.. \quad (PW**0.60)*(PZ**0.40) \quad =G= \quad PY;$$

$$PRICEW.. \quad (E**0.5)*(PY**0.5) \quad =G= \quad PU;$$

$$INDEX.. \quad E \quad =E= \quad (N*PX**(1-EP))**(1/(1-EP));$$

$$INCOME.. \quad CONS \quad =E= \quad PZ*ENDOWS + PW*ENDOWL;$$

$$SKLAB.. \quad ENDOWS \quad =E= \quad 0.40*(PW**0.60)*(PZ**(0.40-1))*Y*100 \\ + 0.6*(PW**0.4)*(PZ**(0.6-1))*N*X*80 \\ + 0.6*(PW**0.4)*(PZ**(0.6-1))*N*FC;$$

$$UNLAB.. \quad ENDOWL \quad =E= \quad 0.60*(PW**(0.60-1))*(PZ**0.40)*Y*100 \\ + 0.4*(PW**(0.4-1))*(PZ**0.6)*N*X*80 \\ + 0.4*(PW**(0.4-1))*(PZ**0.6)*N*FC;$$

MODEL M62 /PRICEX.X, PRICEY.Y, PRICEW.W, ZEROP.N,  
 DX.PX, DY.PY, DW.PU,  
 SKLAB.PZ, UNLAB.PW, INDEX.E, INCOME.CONS/;

OPTION MCP=MILES;

```

OPTION LIMROW=0;
OPTION LIMCOL=0;
$OFFSYMLIST OFFSYMXREF OFFUELLIST OFFFUELXREF

E.L = 1.25;
CONS.L = 200;
X.L = 1;
Y.L = 1;
N.L = 1;
W.L = 1;
PX.L = 1.25;
PY.L = 1;
PZ.L = 1;
PW.L = 1;
PU.L = 1.25**0.5;

PY.FX = 1;

SOLVE M62 USING MCP;
MODELSTAT = M62.MODELSTAT - 1.;

DISPLAY MODELSTAT;

*      Counterfactual: expand the size of the economy

ENDOWS = 200;
ENDOWL = 200;

SOLVE M62 USING MCP;

```

The counter-factual experiment doubles the size of the economy. Note that the results are exactly the same as in the external-economies model M61 (although remember that here  $X$  is output per firm whereas earlier it referred to total output). These models are in fact operationally identical, as I showed in my 1990 CJE article referenced earlier. In both cases, the  $X$  sector's output is homogeneous of degree 1.25 in factor inputs, if by  $X$  sector's output here we mean  $X_c$ . The  $X$  sector expands only through the entry of new firms and  $X_c$  is given by

$$X_c = n^{1/\alpha} X \text{ and in the external economies model by } X = F(V)^{\frac{1}{1-\beta}}$$

where  $X$  in the first equation is a constant. Thus  $\alpha = 0.8$  ( $s = 5$ ) in this model is exactly equivalent to the external economies model with  $\beta = 0.2$ .

Model 62-MPS Closed economy large-group monopolistic competition, MPS/GE version

In the case of "large-group monopolistic competition" where markups are assumed to be fixed, we have seen that firms produce at a fixed scale if variable and fixed costs use factors in the same proportion (this point is almost never recognized in the literature). So we could view the industry production function as producing at constant scale, adding new goods instead of more of existing goods. But then we run into trouble modelling preferences, which must have constant returns to scale in MPS/GE. Doubling industry output means more than doubling utility taking into account the value of increased product diversity.

This is the same problem that we ran into with the external economies model M61 in trying to code it into MPS/GE. We will get around this problem by constructing the MPS/GE model using the tricks of the oligopoly model with free entry of chapter 5 and the external economies mode just mentioned. First, there is a "dummy" good called CX, which is produced with constant marginal cost and a markup is assigned to entrepreneurs just as in the oligopoly model. These entrepreneurs "demand" fixed costs just like in the oligopoly model, and the activity level of fixed costs (N) is interpreted as the number of firms active in equilibrium.

In the "large-group" monopolistic-competition case, the markup is given by  $1/\sigma$  where  $\sigma$  is the elasticity of substitution among the differentiated goods. In the present example, this elasticity is equal to 5, so the markup is 0.20, and we just treat this as a tax on X production with the revenue assigned to the representative agent ENTRE. This agent demands fixed costs as in our earlier oligopoly model and the activity level for the production of fixed costs is interpreted as the number of firms active in equilibrium (units will be chosen such that  $N = 1$  initially; we did not do this in the oligopoly model because the number of firms appears in the markup formula).

But then we have to deal with the consumption side, and this is dealt with in the same fashion as in the external economies case. Let  $X = NX_i$  where N is the number of firms (products) and  $X_i$  is the output per firm.  $X_c$  is defined as in the MCP model.

$$(23) \quad X_c = \left[ \sum X_i^\alpha \right]^{1/\alpha} = \left[ NX_i^\alpha \right]^{1/\alpha} = N^{1/\alpha} X_i = N^{(1-\alpha)/\alpha} NX_i = N^{(1-\alpha)/\alpha} X$$

Now we can use the trick from the external economies model. The X industry produces  $X = (NX_i)$ , but consumers receive  $X_c$ . So we can "give" the consumer (expand the consumer's endowment by the amount):

$$(24) \quad XQADJ = N^{(1-\alpha)/\alpha} X$$

so that the consumer will receive the correct amount of utility from X and

therefore demand the correct amount of X at equilibrium prices.

Finally, we have the same problem as in the external economies case, we must have the value of X received by the consumer equal to the payments received by the producer. Therefore, the consumer has to subsidize the producer of X so that producer revenue equals the payments for X made by the producer.

As in the external economies model, let q be the consumer price, and p be the producer price. We must have

$$q[N^{(1-\alpha)/\alpha}X] = pX = q(1-s)X$$

So we must have

$$s = N^{(1-\alpha)/\alpha} - 1$$

This is given by the endogenous tax rate XPADJ (in the N: field) in the model to follow. Since  $\alpha = 0.8$ ,  $\sigma = 5$ , the value of s is  $s = .25 = (1 - \alpha)/\alpha$ .

To avoid having an ad valorem subsidy multiplied on top of an ad valorem tax (the markup in activity XI), we just specify another activity, simply called X. X produces one unit of "final good" PX for each unit of "intermediate good" "CX". PX is the good that enters welfare and demand.

The counter-factual experiment in the program that follows doubles the size of the economy. The activity levels for X, XI, and N all double, but welfare more than doubles, reflecting the increased value of product diversity ( $W = 2.18$ ). For comparison to the MCP version of this model, I also calculate the price index, e, for  $X_c$  which is generated after the model solves. This is declared as the parameter INDEX.

\$TITLE: Model M62-MPS: Large-Group Monopolistic Competition, uses MPS/GE

\$ONTEXT

Markets	Production Sectors				Consumers		ENTR
	XI	X	N	Y	W	CONS	
PX		100			-100		
CX	100	-100					
PY				100	-100		
PF			20				-20
PU					200	-200	
PW	-32		-8	-60		100	
PZ	-48		-12	-40		100	
MK	-20						20

\$OFFTEXT

PARAMETERS

ENDOW Size index for the economy  
 INDEX Price index for the X goods  
 EP Elasticity of substitution among X varieties;

ENDOW = 1;

EP = 5;

\$ONTEXT

\$MODEL:M62

\$SECTORS:

X ! Activity level for sector X  
 Y ! Activity level for sector Y  
 W ! Activity level for sector W (Hicksian welfare index)  
 N ! Activity level for sector X fixed costs, no. of firms  
 XI ! Activity level -- marginal cost of X

\$COMMODITIES:

PX ! Price index for commodity X (gross of markup)  
 CX ! Marginal cost index for commodity X (net markup)  
 PY ! Price index for commodity Y  
 PW ! Price index for unskilled labor  
 PZ ! Price index for skilled labor  
 PF ! Unit price of inputs to fixed cost  
 PU ! Price index for welfare (expenditure function)

## \$CONSUMERS:

CONS ! Income level for consumer CONS  
 ENTRE ! Entrepreneur (converts markup revenue to fixed cost)

## \$AUXILIARY:

XQADJ ! Quantity adjustment (positive when X>1)  
 XPADJ ! X output subsidy rate (positive when X>1)

## \$PROD:X s:1

O:PX Q: 80 P:1.25 A:CONS N:XPADJ M:-1  
 I:CX Q: 80 P:1.25

## \$PROD:Y s:1

O:PY Q:100  
 I:PW Q: 60  
 I:PZ Q: 40

## \$PROD:XI s:1

O:CX Q: 80 A:ENTRE T:0.20  
 I:PW Q: 32  
 I:PZ Q: 48

## \$PROD:N s:1

O:PF Q:20  
 I:PZ Q:12  
 I:PW Q: 8

## \$PROD:W s:1.0

O:PU Q:200  
 I:PX Q: 80 P:1.25  
 I:PY Q:100

## \$DEMAND:CONS

D:PU Q:200  
 E:PW Q:(100\*ENDOW)  
 E:PZ Q:(100\*ENDOW)  
 E:PX Q:80 R:XQADJ

## \$DEMAND: ENTRE

D:PF Q:20

## \$CONSTRAINT:XQADJ

XQADJ =E= (N\*\*(1/(EP-1)))\*X - X;

## \$CONSTRAINT:XPADJ

XPADJ =E= (N\*\*(1/(EP-1))) - 1;

```
$OFFTEXT
$SYSINCLUDE mpsgeset M62

*       Adjust bounds so that the auxiliary variables can take on
*       negative values:

XQADJ.LO = -INF;
XPADJ.LO = -INF;

*       Benchmark replication:
PY.FX = 1;
PX.L = 1.25;
CX.L = 1.25;

$INCLUDE M62.GEN
SOLVE M62 USING MCP;

INDEX = (N.L*CX.L**(1-EP))**(1/(1-EP));
DISPLAY INDEX;

*       Counterfactual: expand the size of the economy

ENDOW = 2;

$INCLUDE M62.GEN
SOLVE M62 USING MCP;

INDEX = (N.L*CX.L**(1-EP))**(1/(1-EP));
DISPLAY INDEX;
```

Model M63-MCP Two-country model with large-group monopolistic competition, MCP version

Now I present a two-country version of the monopolistic-competition model. This model has been popular in trade theory and in the so-called new economic geography literature. The countries will be labeled with subscripts  $i$  and  $j$ .  $X$ 's will carry double subscripts where the first is the country of production and the second is the country of consumption. Thus  $X_{ij}$  is  $X$  produced in  $i$  and exported to  $j$  where it is consumed.

Trade costs are important in the literatures just referred to, so we will want to consider these in our model. It has been popular to use "iceberg" trade costs where some of the good "melts" in transit. The first thing to note is how iceberg trade costs are reflected in prices and demand. For a domestic firm,  $X_{ij}$  is the amount produced in country  $i$  and shipped to country  $j$ . Let  $t$  ( $t > 1$ ) be the ratio of the amount of  $X$  exported to the amount that arrives "unmelted". Alternatively  $1/t$  is the proportion of a good that "survives" transit (the proportion "unmelted"). If  $X_{ij}$  is shipped, the amount received in country  $j$  is  $X_{ij}/t$ .

Second, we make the usual assumption that there is no price discrimination and so the home price of a good for local sales equals its export price. Thus we can use the notation  $p_i$  and  $p_j$  for the price of all goods *produced* in country  $i$  and country  $j$  respectively. The revenues received by the exporter are equal to the costs paid by the importer:  $p_i X_{ij}$  is the revenue received by the exporter and  $X_{ij}/t$  are the number of units arriving in the importing country, so the price per unit in the importing country must be  $p_j t$  ( $p_i X_{ij} = (p_j t) X_{ij}/t$ ). Rather than introduce additional notation, we will therefore use  $X_{ij}/t$  and  $p_j t$  as the quantity and price in country  $j$  of a country  $i$  variety exported to country  $j$ .

The price index for country  $i$  is then given by:

$$e_i = \left[ N_i p_i^{1-\sigma} + N_j (p_j t)^{1-\sigma} \right]^{\frac{1}{1-\sigma}} \quad (15)$$

The various  $X$  varieties sold in country  $i$  are given by:

$$X_{ii} = p_i^{-\sigma} e_i^{\sigma-1} M_{ix} \quad X_{ji}/t = (p_j t)^{-\sigma} e_i^{\sigma-1} M_{ix} \quad (16)$$

where the second equation can also be written as:

$$X_{ji} = p_j^{-\sigma} t^{1-\sigma} e_i^{\sigma-1} M_{ix} \quad (17)$$

There are few other complications beyond those discussed in the closed economy model

M62-MCP. The model is calibrated to trade free with the countries identical initially. Since X is differentiated, it must therefore be the case that half of each firm's output is sold in its home country and half is exported. There are some numerical coefficients in the problem used to adjust units so that most prices are one initially; this is not necessary but makes the output file more comparable to the MPS/GE version that follows. This arises because I set the number of firms in each country to 1 initially, so that the total number is 2. But if the price of an individual variety is 1, then the price index in each country cannot be equal to 1. I thus use a units adjustment in the equations for WELFAREI, WELFAREJ, PUI and PUJ in order that these all take the value of 1 initially.

Several counterfactual experiments are considered at the end. First, we double the size of the economies and see that there welfare more than doubles. In the second experiment, we return the economy to its original size but set the trade cost at  $TC = 1.15$ . This reduces welfare by 3% in each country as consumption of home versus imported goods becomes more unbalanced. The third experiment has country i twice as large as country j, but no trade costs. Welfare is the same for both countries (per capita) in free trade.

The fourth experiment has country i three times the size of j, but with trade costs  $TC = 1.15$ . Here we see some results found in the newer theory literature. There is a "home market effect" whereby country i is better off and has more firms on a per capita basis than country j. At the end of this experiment, we calculate real factor prices in each country by dividing the nominal prices by the real consumer prices indices, the cost of buying one unit of utility. We see that in spite of the countries having identical relative endowments, country i has a higher real price for skilled labor and country j has a higher real price for unskilled labor. Thus this equilibrium is not going to be stable under factor mobility, and skilled labor should flow toward country i increasing further its specialization in X.

The final experiment retains the trade costs, gives both countries the same amount of unskilled labor, but country i has 60% of the skilled labor and country j 40%. If you look at the very end of the listing file, you see that country i has higher real prices for both factors, an outcome noted by Markusen and Venables (JIE 2000).

\$TITLE: Model M63: Two Country Large-Group Monopolistic Competition  
 \* uses MCP

\$ontext

	YI	YJ	XMI	XMJ	NMI	NMJ	WI	WJ	CONI	CONJ	EHTI	ENTJ
PYI	100						-100					
PYJ		100						-100				
PXI			100				-50	-50				
PXJ				100			-50	-50				
FCI					20						-20	
FCJ						20						-20
ZI	-40		-48		-12				100			
ZJ		-40		-48		-12				100		
WI	-60		-32		-8				100			
WJ		-60		-32		-8				100		
UTILI							200		-200			
UTILJ								200		-200		
MKI			-10	-10							10	10
MKJ			-10	-10							10	10

\$offtext

PARAMETERS

EP

EY

TC

FC

ENDOWIS

ENDOWIL

ENDOWJS

ENDOWJL

MODELSTAT;

EP = 5;

EY = 3;

TC = 1.;

FC = 20;

ENDOWIS = 100;

ENDOWIL = 100;

ENDOWJS = 100;

ENDOWJL = 100;

POSITIVE VARIABLES

WFI

WFJ  
XII  
XIJ  
XJJ  
XJI  
YI  
YJ  
NI  
NJ  
PI  
PJ  
PY  
PUI  
PUJ  
EI  
EJ  
ZI  
WI  
ZJ  
WJ  
MI  
MJ;

EQUATIONS  
WELFAREI  
WELFAREJ  
DXII  
DXJI  
DXJJ  
DXIJ  
DY  
ZEROPI  
ZEROPJ  
PRICEI  
PRICEJ  
PRICYI  
PRICYJ  
PRICEUI  
PRICEUJ  
DWJ  
INDEXI  
INDEXJ  
EXPI  
EXPJ  
SKLABI  
UNLABI  
SKLABJ  
UNLABJ;

WELFAREI..  $200*WFI =E= ((2**(1/(1-EP)))*1.25)**0.5*MI/(1.025*PUI);$   
 WELFAREJ..  $200*WFJ =E= ((2**(1/(1-EP)))*1.25)**0.5*MJ/(1.025*PUJ);$   
 DXII..  $XII*40 =E= PI**(-EP)*(EI**(EP-1))*MI/2;$   
 DXJI..  $XJI*40/TC =E= (PJ*TC)**(-EP)*(EI**(EP-1))*MI/2;$   
 DXJJ..  $XJJ*40 =E= PJ**(-EP)*(EJ**(EP-1))*MJ/2;$   
 DXIJ..  $XIJ*40/TC =E= (PI*TC)**(-EP)*(EJ**(EP-1))*MJ/2;$   
 DY..  $YI*100 + YJ*100 =E= MI/(2*PY) + MJ/(2*PY);$   
 ZEROPI..  $FC*(EP-1) =G= XII*40 + XIJ*40;$   
 ZEROPJ..  $FC*(EP-1) =G= XJJ*40 + XJI*40;$   
 PRICEI..  $(WI**0.4)*(ZI**0.6) =G= PI*(1-1/EP);$   
 PRICEJ..  $(WJ**0.4)*(ZJ**0.6) =G= PJ*(1-1/EP);$   
 PRICYI..  $(WI**0.60)*(ZI**0.40) =G= PY;$   
 PRICYJ..  $(WJ**0.60)*(ZJ**0.40) =G= PY;$   
 PRICEUI..  $(EI**0.5)*(PY**0.5)/1.025 =G= PUI;$   
 PRICEUJ..  $(EJ**0.5)*(PY**0.5)/1.025 =G= PUJ;$   
 INDEXI..  $EI =E= (NI*PI**(1-EP) + NJ*(PJ*TC)**(1-EP))**(1/(1-EP));$   
 INDEXJ..  $EJ =E= (NI*(PI*TC)**(1-EP) + NJ*PJ**(1-EP))**(1/(1-EP));$   
 EXPI..  $MI =E= ZI*ENDOWIS + WI*ENDOWIL;$   
 EXPJ..  $MJ =E= ZJ*ENDOWJS + WJ*ENDOWJL;$   
  
 SKLABI..  $ENDOWIS =E= 0.40*(WI**0.60)*(ZI**(0.40-1))*YI*100$   
 $+ 0.6*(WI**0.4)*(ZI**(0.6-1))*NI*((XII+XIJ)*40 + FC);$   
  
 UNLABI..  $ENDOWIL =E= 0.60*(WI**(0.60-1))*(ZI**0.40)*YI*100$   
 $+ 0.4*(WI**(0.4-1))*(ZI**0.6)*NI*((XII+XIJ)*40 + FC);$

SKLABJ..      ENDOWJS =E= 0.40\*(WJ\*\*0.60)\*(ZJ\*\*(0.40-1))\*YJ\*100  
                   + 0.6\*(WJ\*\*0.4)\*(ZJ\*\*(0.6-1))\*NJ\*( (XJJ+XJI)\*40 + FC);

UNLABJ..      ENDOWJL =E= 0.60\*(WJ\*\*(0.60-1))\*(ZJ\*\*0.40)\*YJ\*100  
                   + 0.4\*(WJ\*\*(0.4-1))\*(ZJ\*\*0.6)\*NJ\*( (XJJ+XJI)\*40 + FC);

MODEL M63 /    WELFAREI.WFI, WELFAREJ.WFJ,  
                   PRICYI.YI, PRICYJ.YJ, DXII.XII, DXJI.XJI,  
                   DXJJ.XJJ, DXIJ.XIJ, DY.PY,  
                   ZEROPI.NI, ZEROPJ.NJ,  
                   PRICEI.PI, PRICEJ.PJ,  
                   PRICEUI.PUI, PRICEUJ.PUJ,  
                   SKLABI.ZI, SKLABJ.ZJ, UNLABI.WI,  
                   UNLABJ.WJ, INDEXI.EI, INDEXJ.EJ, EXPI.MI, EXPJ.MJ/;

OPTION MCP=MILES;  
 OPTION LIMROW=0;  
 OPTION LIMCOL=0;  
 \$OFFSYMLIST OFFSYMREF OFFUELLIST OFFFUELXREF

WFI.L = 1;  
 WFJ.L = 1;  
 PUI.L = 1.25\*\*0.5;  
 PUJ.L = 1.25\*\*0.5;  
 EI.L = 1;  
 EJ.L = 1;  
 MI.L = 200;  
 MJ.L = 200;  
 XII.L = 1;  
 XIJ.L = 1;  
 XJJ.L = 1;  
 XJI.L = 1;  
 YI.L = 1;  
 YJ.L = 1;  
 NI.L = 1;  
 NJ.L = 1;  
 PI.L = 1.25;  
 PJ.L = 1.25;  
 PY.L = 1;  
 ZI.L = 1;  
 WI.L = 1;  
 ZJ.L = 1;  
 WJ.L = 1;

PY.FX = 1;

TC = 1.;

SOLVE M63 USING MCP;  
MODELSTAT = M63.MODELSTAT - 1.;

ENDOWIS = 200;  
ENDOWIL = 200;  
ENDOWJS = 200;  
ENDOWJL = 200;

SOLVE M63 USING MCP;

TC = 1.15;

ENDOWIS = 100;  
ENDOWIL = 100;  
ENDOWJS = 100;  
ENDOWJL = 100;

SOLVE M63 USING MCP;

TC = 1.0;

ENDOWIS = 150;  
ENDOWIL = 150;  
ENDOWJS = 50;  
ENDOWJL = 50;

SOLVE M63 USING MCP;

TC = 1.15;

ENDOWIS = 150;  
ENDOWIL = 150;  
ENDOWJS = 50;  
ENDOWJL = 50;

SOLVE M63 USING MCP;

REALWI = WI.L/PUI.L;  
REALWJ = WJ.L/PUJ.L;  
REALZI = ZI.L/PUI.L;  
REALZJ = ZJ.L/PUJ.L;

```
DISPLAY REALWI, REALWJ, REALZI, REALZJ;
```

```
ENDOWIS = 120;  
ENDOWIL = 100;  
ENDOWJS = 80;  
ENDOWJL = 100;
```

```
SOLVE M63 USING MCP;
```

```
REALWI = WI.L/PUI.L;  
REALWJ = WJ.L/PUJ.L;  
REALZI = ZI.L/PUI.L;  
REALZJ = ZJ.L/PUJ.L;
```

```
DISPLAY REALWI, REALWJ, REALZI, REALZJ;
```

Model M63-MPS Two-country model with large-group monopolistic competition, MPS/GE version

This is the MPS/GE version of the same two-country monopolistic-competition model. At this point, I don't think it requires much additional explanation. Note how iceberg trade costs conveniently fit into the production blocks for export sales.

```
$PROD: XIJ
O: PXIJ   Q: (40./TC)   A: CONSJ   N: SUBI   M: -1.0
I: PXI    Q: (40.)
```

If 40 units are sent out, then  $40/TC$  arrive unmelted.

It is also important to remember this in the XQADJ constraints.  $X_{ij}$  is the amount exported from  $i$  under iceberg trade costs, but only  $X_{ij}/TC$  is received in country  $j$ .

```
$CONSTRAINT: XQADJIJ
XQADJIJ =E= (NI**(.25))*XIJ/TC - XIJ/TC;
```

\$TITLE: Model M63: Two Country Large-Group Monopolistic Competition  
 \* uses MPS/GE

\$ontext

	YI	YJ	XI	XJ	NI	NJ	WI	WJ	CONI	CONJ	EHTI	ENTJ
PYI	100						-100					
PYJ		100						-100				
PXI			100				-50	-50				
PXJ				100			-50	-50				
FCI					20						-20	
FCJ						20						-20
ZI	-40		-48		-12				100			
ZJ		-40		-48		-12				100		
WI	-60		-32		-8				100			
WJ		-60		-32		-8				100		
UTILI							200		-200			
UTILJ								200		-200		
MKI			-10	-10							10	10
MKJ			-10	-10							10	10

\$offtext

PARAMETERS

ENDOWIL  
 ENDOWJL  
 ENDOWIS  
 ENDOWJS  
 TC  
 SUBSIDY  
 SIGMA  
 REALWI  
 REALWJ  
 REALZI  
 REALZJ;

ENDOWIL = 1.;  
 ENDOWJL = 1.;  
 ENDOWIS = 1.;  
 ENDOWJS = 1.;  
 TC = 1;  
 SUBSIDY = 0;  
 SIGMA = 5;

\$ONTEXT  
\$MODEL:M63

\$SECTORS:  
WFI WFJ  
XI  
XII  
XIJ  
XJ  
XJI  
XJJ  
YI YJ  
NI NJ

\$COMMODITIES:  
PY  
PUI PUJ  
WI WJ  
ZI ZJ  
PXII PXJI PXJJ PXIJ  
PXI  
PXJ  
FCI FCJ

\$CONSUMERS:  
CONSI CONSJ  
ENTI ENTJ

\$AUXILIARY:  
XQADJII  
XQADJJI  
XQADJJJ  
XQADJIJ  
XPADJI  
XPADJJ

\$PROD:YI s:1.0  
O:PY Q:100.0  
I:WI Q: 60.0  
I:ZI Q: 40.0

\$PROD:YJ s:1.0  
O:PY Q:100.0  
I:WJ Q: 60.0  
I:ZJ Q: 40.0

\$PROD:XI s:1.0

O:PXI Q:80. A:ENTI T:.20  
 I:ZI Q:48  
 I:WI Q:32

## \$PROD:XII

O:PXII Q:40. A:CONSI N:XPADJI M:-1.0  
 I:PXI Q:40.

## \$PROD:XIJ

O:PXIJ Q:(40./TC) A:CONSJ N:XPADJI M:-1.0  
 I:PXI Q:(40.)

## \$PROD:XJ s:1.0

O:PXJ Q:80. A:ENTJ T:.20  
 I:ZJ Q:48  
 I:WJ Q:32

## \$PROD:XJI

O:PXJI Q:(40./TC) A:CONSI N:XPADJJ M:-1.0  
 I:PXJ Q:(40.)

## \$PROD:XJJ

O:PXJJ Q:40. A:CONSJ N:XPADJJ M:-1.0  
 I:PXJ Q:40.

## \$PROD:NI s:1.0

O:FCI Q:20  
 I:ZI Q:12  
 I:WI Q:8

## \$PROD:NJ s:1.0

O:FCJ Q:20  
 I:ZJ Q:12  
 I:WJ Q:8

## \$PROD:WFI s:1.0 a:5.0

O:PUI Q:200.  
 I:PXII Q:40. P:1.25 a:  
 I:PXJI Q:40 P:1.25 a:  
 I:PY Q:100.

## \$PROD:WFJ s:1.0 a:5.0

O:PUJ Q:200.  
 I:PXJJ Q:40. P:1.25 a:  
 I:PXIJ Q:40. P:1.25 a:  
 I:PY Q:100.

\$DEMAND:CONSI

D:PUI

E:WI Q:(100\*ENDOWIL)

E:ZI Q:(100\*ENDOWIS)

E:PXII Q:40 R:XQADJII

E:PXJI Q:40 R:XQADJJI

\$DEMAND:CONSJ

D:PUJ

E:WJ Q:(100\*ENDOWJL)

E:ZJ Q:(100\*ENDOWJS)

E:PXIJ Q:40 R:XQADJIJ

E:PXJJ Q:40 R:XQADJJI

\$DEMAND:ENTI

D:FCI

\$DEMAND:ENTJ

D:FCJ

\$CONSTRAINT:XQADJII

XQADJII =E= (NI\*\*(.25))\*XII - XII;

\$CONSTRAINT:XQADJJI

XQADJJI =E= (NJ\*\*(.25))\*XJI/TC - XJI/TC;

\$CONSTRAINT:XQADJJI

XQADJJI =E= (NJ\*\*(.25))\*XJJ - XJJ;

\$CONSTRAINT:XQADJIJ

XQADJIJ =E= (NI\*\*(.25))\*XIJ/TC - XIJ/TC;

\$CONSTRAINT:XPADJI

XPADJI =E= NI\*\*.25 - 1;

\$CONSTRAINT:XPADJJ

XPADJJ =E= NJ\*\*.25 - 1;

\$OFFTEXT

\$SYSINCLUDE MPSGESET M63

XQADJII.LO = -INF;

XQADJIJ.LO = -INF;

XQADJJI.LO = -INF;

XQADJJI.LO = -INF;

```
XPADJI.LO = -INF;  
XPADJJ.LO = -INF;
```

```
PY.FX = 1;
```

```
PXII.L = 1.25;  
PXIJ.L = 1.25;  
PXJJ.L = 1.25;  
PXJI.L = 1.25;  
PXI.L = 1.25;  
PXJ.L = 1.25;  
XII.L = 1;  
XIJ.L = 1;  
XJJ.L = 1;  
XJI.L = 1;
```

```
M63.ITERLIM = 5000;  
OPTION LIMROW=0;  
OPTION LIMCOL=0;  
$OFFSYMLIST OFFSYMREF OFFUELLIST OFFFUELXREF
```

```
$INCLUDE M63.GEN  
SOLVE M63 USING MCP;
```

```
*counterfactual: double size of world economy
```

```
ENDOWIL = 2;  
ENDOWJL = 2;  
ENDOWIS = 2;  
ENDOWJS = 2;
```

```
$INCLUDE M63.GEN  
SOLVE M63 USING MCP;
```

```
TC = 1.15;
```

```
ENDOWIL = 1;  
ENDOWJL = 1;  
ENDOWIS = 1;  
ENDOWJS = 1;
```

```
$INCLUDE M63.GEN  
SOLVE M63 USING MCP;
```

```
TC = 1.0;
```

```
ENDOWIL = 1.5;  
ENDOWJL = .5;  
ENDOWIS = 1.5;  
ENDOWJS = .5;
```

```
$INCLUDE M63.GEN  
SOLVE M63 USING MCP;
```

```
TC = 1.15;
```

```
ENDOWIL = 1.5;  
ENDOWIS = 1.5;  
ENDOWJL = 0.5;  
ENDOWJS = 0.5;
```

```
$INCLUDE M63.GEN  
SOLVE M63 USING MCP;
```

```
REALWI = WI.L/PUI.L;  
REALWJ = WJ.L/PUJ.L;  
REALZI = ZI.L/PUI.L;  
REALZJ = ZJ.L/PUJ.L;
```

```
DISPLAY REALWI, REALWJ, REALZI, REALZJ;
```

```
ENDOWIL = 1.0;  
ENDOWIS = 1.2;  
ENDOWJL = 1.0;  
ENDOWJS = 0.8;
```

```
$INCLUDE M63.GEN  
SOLVE M63 USING MCP;
```

```
REALWI = WI.L/PUI.L;  
REALWJ = WJ.L/PUJ.L;  
REALZI = ZI.L/PUI.L;  
REALZJ = ZJ.L/PUJ.L;
```

```
DISPLAY REALWI, REALWJ, REALZI, REALZJ;
```